
	Reason for change:
	In current TTCN Implementation of the function f_TC_8_1_5_8_1_NR5GC_TestBody,

1. At Step 14a5, while transmitting RRC Resume Message to the UE , 
ReestablishPDCP IE is not set for SRB2 and DRB. It is not in align with Table 4.6.3-132: RadioBearerConfig from 38508

2. At Step 23, 1s delay is in place after the radio link quality of NR Cell 1 is degraded . before receiving the reestablishment request , there are local configurations in place. This delay leads to stability issues in the test case ,  that reestablishment request is received in the middle of local configurations 

In the implementation of function f_NR5GC_RRC_ProcedureDelay and in f_NR5GC_RRC_ProcedureDelayReestablishment
3.Disabling of UL HARQ System Indications are done once HARQ ACK is received . there are possibilities of race condition where TTCN is waiting for  the CNF for Disable UL HARQ System indication , but ended up receiving SRB COMMON IND. 
And In case of Reestablishment complete triggered based on Rach Procedure , Disabling Prach System Indications are done once Prach Preamble is received . There are possibilities of race condition where TTCN is waiting for CNF for disable Prach System Indication , but ended up receiving SRB COMMON IND. This race condition has to be addressed

In the function f_NR5GC_RRC_ProcedureDelayReestablishment,
4. when UE has triggered RACH procedure for sending the reestablishment complete message , “repeat” is missing inside the alt block . this results in SRB COMMON IND ( Reestablishment complete ) reception failure

5. v_T1Slots calculcation is not considering v_MaxNumberSlots

	
	

	Summary of change:
	In the function function f_TC_8_1_5_8_1_NR5GC_TestBody:
1. At Step 14a5 ReestablishPDCP IE is set to ‘TRUE’ while sending the RRC Resume Message

2. At Step 23 , instead of having 1s delay function call , it is proposed to have the timer start function call after the radio link quality of NR Cell 1 is degraded . 

In the function f_NR5GC_RRC_ProcedureDelay and in f_NR5GC_RRC_ProcedureDelayReestablishment
3.To avoid the race condition between the reception of SYSTEM CONTROL CNF for Disbaling UL HARQ indications and SRB COMMON IND , new handler is added inside the alt block.
Race condition between the reception of SYSTEM CONTROL CNF for Disabling Prach Preamble Indications and SRB COMMON IND is taken care as well ; as an additional sanity check , when receiving SRB COMMON IND , new check is added to make sure that UL HARQ IND is already received

In the function f_NR5GC_RRC_ProcedureDelayReestablishment
4.repeat statement is added inside alt block , when receiving the Rach Preamble Indication , so that Reestablishment complete gets handled inside the same alt block

5.MaxNumberSlots is taken into consideration while calculating  v_T1Slots

	
	

	Consequences if not approved:
	A conformant UE may fail the test case

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc103669274]Corrections required
0.1 [bookmark: _Toc103669275]Correction to the function f_TC_8_1_5_8_1_NR5GC_TestBody
	Function name
	f_TC_8_1_5_8_1_NR5GC_TestBody

	Reason for change
	In current TTCN Implementation of the function f_TC_8_1_5_8_1_NR5GC_TestBody,

1. At Step 14a5, while transmitting RRC Resume Message to the UE , 
ReestablishPDCP IE is not set for SRB2 and DRB. It is not in align with Table 4.6.3-132: RadioBearerConfig from 38508

2. At Step 23, 1s delay is in place after the radio link quality of NR Cell 1 is degraded . before receiving the reestablishment request , there are local configurations in place. This delay leads to stability issues in the test case ,  that reestablishment request is received in the middle of local configurations 

	Summary of change
	In the function function f_TC_8_1_5_8_1_NR5GC_TestBody:
1. At Step 14a5 ReestablishPDCP IE is set to ‘TRUE’ while sending the RRC Resume Message

2. At Step 23 , instead of having 1s delay function call , it is proposed to have the timer start function call after the radio link quality of NR Cell 1 is degraded

	TTCN module
	develop\NR5GC\8_1_5\RRC_Others_NR5GC.ttcn

	MCC160 Comment
	



    
Before Change:
	  function f_TC_8_1_5_8_1_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var RNTI_Value v_RNTI_Value;
    var PhysCellId v_PhysCellId;
    var ShortMAC_I v_ShortMAC_I;
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters;
    var NR_SecurityParams_Type v_SecurityParams;
    var octetstring v_EncodeCapRequestNR;
    var template (value) DRB_CountMSB_InfoList v_CountList;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) CellGroupConfig v_CellGroupConfig;
    var NR_RadioBearerList_Type v_SS_Config;
    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var template (value) NR_RRC_MSG_Request_Type v_RrcMsg;
    var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
    var template(omit) octetstring v_EncodedSCG_Config := omit;
    var NR_DRBInfoList_Type v_ExistingDRBs;
    var template(value) NR_RachProcedureConfig_Type v_NR_RachProcedureConfig;
    var integer v_Max_nPRB := f_NR_CellInfo_GetCarrierBandwidth_DL(nr_Cell1);
    var NR_ResourceAllocation_Type v_NR_ResourceAllocation := {v_Max_nPRB, 28};
    var template DRB_CountInfoList v_CountList_UE;
    var template (value) NG_NAS_MSG_RequestList_Type v_ServerAcceptMsg;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;

    v_CellPowerList_AtT0 :={
      cs_NR_CellPower(nr_Cell2, -96, -96) //FFS in prose for FR2
    };
    v_CellPowerList_AtT1:={
      cs_NR_CellPower(nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),   //FFS in prose for FR2
      cs_NR_CellPower(nr_Cell2, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)          //FFS in prose for FR2
    };
    f_NR_SetCellPowerList(v_CellPowerList_AtT0);

    //@siclog "Step 1" siclog@
    //The SS transmits a Paging message including a matched ng-5G-S-TMSI.
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);

    //@siclog "Step 2" siclog@
    //The UE transmits an RRCSetupRequest message.
    f_NR_RRC_SetupRequest_Def(nr_Cell1);

    //@siclog "Steps 3 - 4" siclog@
    //The SS transmits an RRCSetup message.
    //The SS starts transmitting a periodic DCI from (10*2^numerology+1+TDL) slots after step 3 to schedule PUSCH.
    //Check: Does the UE transmit an RRCSetupComplete message within (10*2^numerology+1+TDL+K2) slots15 (FDD)/18 (TDD) subframes after successful completion of step 3?
    v_ReceivedNR_SRB_COMMON_IND := f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                                                              cas_NR_SRB0_RrcPdu_REQ(nr_Cell1,cs_TimingInfo_Now,f_NR_38508_RRCSetup(nr_Cell1, tsc_NR_RRC_TI_Def)),
                                                              car_NR_SRB1_RrcNasPdu_IND(nr_Cell1, cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *), cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected)),
                                                              10, "Step 4" );

    //@siclog "Step 5 - 6" siclog@
    //The SS transmits a SecurityModeCommand message.
    //The SS starts transmitting a periodic DCI from (5*2^numerology+1+TDL) slots after step 5 to schedule PUSCH.
    //Check: Does the UE transmit a SecurityModeComplete message within (5*2^numerology+1+TDL+K2) 10 (FDD)/13 (TDD) subframes slots after successful completion of step 5?
    v_SecurityParams := f_NR_Security_Get();
    v_SecurityParams := f_NR_Authentication_InitAS (v_SecurityParams, v_ReceivedNR_SRB_COMMON_IND.Signalling.Nas[0].SecurityProtection.NasCount);
    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(nr_Cell1);
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(nr_Cell1,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);

    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,
                                                      -,
                                                      cs_38508_RRCSecurityModeCommand(tsc_NR_RRC_TI_Def,
                                                                                      cs_38508_SecurityAlgorithmConfig(v_SecurityParams.AS_Ciphering.Algorithm,
                                                                                                                       v_SecurityParams.AS_Integrity.Algorithm))),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCSecurityModeComplete(tsc_NR_RRC_TI_Def)),
                               5, "Step 6" );

    f_NR_Security_Set(v_SecurityParams);

    //@siclog "Step 7 - 8" siclog@
    //The SS transmits a UECapabilityEnquiry message.
    //The SS starts transmitting a periodic DCI from (80*2^numerology+1+TDL) slots after step 7 to schedule PUSCH.
    //Check: Does the UE transmit a UECapabilityInformation message within (80*2^numerology+1+TDL+K2) slots after successful completion of step 7?
    v_EncodeCapRequestNR := bit2oct(encvalue( cs_38508_UECapabilityRequestFilterNR({ cs_NR_BandInformationNR(f_NR_CellInfo_GetBandIndicator(nr_Cell1)) }) ));
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_UECapabilityEnquiry(tsc_NR_RRC_TI_Def, nr, v_EncodeCapRequestNR)),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_UECapabilityInformationAny(tsc_NR_RRC_TI_Def)),
                               80, "Step 8",
                               cs_NR_FreqDomainSchedulExplicit(-, v_NR_ResourceAllocation.Lrbs)  //DCI with the maximum grant allowed by the carrier bandwidth  @sic R5-217815 sic@
                               );

    //@siclog "Step 9 - 10" siclog@
    //The SS transmits an RRCReconfiguration message to establish SRB2 and DRB #n for the UE
    f_NR_BuildRRCReconfigExistingDRBs (v_RadioBearerConfig, v_CellGroupConfig, v_SS_Config, nr_Cell1);
    
    v_SecurityParams := f_NR_Security_Get();
    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(v_CellGroupConfig)));

    //Reconfigure DRB in SS - existing DRBs are not reconfigured during release
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, v_SS_Config, cs_TimingInfo_Now);
    
    // Configure PDCP security
    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(nr_Cell1, false);
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(nr_Cell1,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);
    v_ServerAcceptMsg := {cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, f_Get_NG_ServiceAcceptMsg(v_ReceivedNR_SRB_COMMON_IND.Signalling.Nas[0].Pdu.Msg.service_Request.pduSessionStatus))};
    v_RrcMsg := cs_NR_RRC_MSG_Request_DCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig, -, -, v_V1530Ext));
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_RrcNasPduList_REQ(nr_Cell1, tsc_NR_RbId_SRB1, cs_TimingInfo_Now, v_RrcMsg, v_ServerAcceptMsg),
                               car_NR_SRB_RrcPdu_IND(nr_Cell1, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete),
                               10, "Step 10" );

    //@siclog "Step 11 - 12" siclog@
    //The SS transmits a CounterCheck message.
    //The SS starts transmitting a periodic DCI from (5*2^numerology+1+TDL) slots after step 11 to schedule PUSCH.
    //Check: Does the UE transmit a CounterCheckResponse message within (5*2^numerology+1+TDL+K2)10 (FDD)/13 (TDD) subframes slots after successful completion of step 11?
    v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();
    v_CountList := {cs_38508_DRB_CountMSB_Info(v_ExistingDRBs[0].DRB_Id, 1, 1)};
    v_CountList_UE := {cr_DRB_CountInfo(v_ExistingDRBs[0].DRB_Id)};
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_CounterCheck(tsc_NR_RRC_TI_Def, v_CountList)),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_CounterCheckResponse(tsc_NR_RRC_TI_Def, v_CountList_UE)),
                               5, "Step 12" );

    //@siclog "Step 13" siclog@
    //Void

    //@siclog "Step 14a1-14a7" siclog@
    if ( pc_inactiveState ){

        //@siclog "Step 14a1" siclog@
        //The SS transmits an RRCRelease message with suspendConfig.
        f_NR_RRC_InactiveDef(nr_Cell1);

        //@siclog "Step 14a2" siclog@
        //Wait 1s to let UE enter RRC_INACTIVE state.
        f_Delay(1.0);

        //@siclog "Step 14a3" siclog@
        //The SS transmits a Paging message including a matched ng-5G-S-TMSI.
        f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);

        v_CellGroupConfig := f_NR_GetSCGConfigResume(nr_Cell1);
        v_EncodedSCG_Config := bit2oct(encvalue(v_CellGroupConfig));

        //@siclog "Step 14a4" siclog@
        //The UE transmits an RRCResumeRequest message
        SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, mt_Access)));

        //@siclog "Step 14a5" siclog@
        //The SS transmits an RRCResume message for the UE

        if(pc_reducedCP_Latency){
            //@siclog "Step 14a5Aa1 - 14a5Aa2" siclog@
            //The SS starts transmitting a periodic DCI from (10* 2 2^numerology +1+TDL) slots after step 14a5 to schedule PUSCH.
            //Check: Does the UE transmit an RRCResumeComplete message within (6*2^numerology+1+TDL+K2) slots after successful completion of step 14a5?
            f_NR5GC_RRC_ProcedureDelay( nr_Cell1,
                                        cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, cs_38508_RRCResume(tsc_NR_RRC_TI_Def, v_EncodedSCG_Config, v_RadioBearerConfig)),
                                        car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeComplete),
                                        6, "Step 14a5Aa2" );
        } else {
            //@siclog "Step 14a5Ab1 - 14a5Ab2" siclog@
            //The SS starts transmitting a periodic DCI from n slots after step 14a5 to schedule PUSCH.
            //Check: Does the UE transmit an RRCResumeComplete message within (10* 2 2^numerology +1+TDL+K2)slots after successful completion of step 14a5?
            f_NR5GC_RRC_ProcedureDelay( nr_Cell1,
                                        cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, cs_38508_RRCResume(tsc_NR_RRC_TI_Def, v_EncodedSCG_Config, v_RadioBearerConfig)),
                                        car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeComplete),
                                        10, "Step 14a5Ab2" );
        }
        v_NR_RachProcedureConfig := f_NR_InitialiseRachProcedure(nr_Cell1);
        f_NR_SS_RachProcedureConfig(nr_Cell1, v_NR_RachProcedureConfig);

        //@siclog "Step 14a6-14a7" siclog@
        //Void
    }
    
    //@siclog "Step 14-21" siclog@
    //Void

    //@siclog "Step 22" siclog@
    //The SS re-adjusts the SS levels according to row "T1" in table 8.1.5.8.1.3.2-1/2 in order that the radio link quality of NR Cell 1 is degraded, and NR Cell 2 is suitable for camping
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

     //@siclog "Step 23" siclog@
    //Wait 1s to let T310 expire
    f_Delay(1.0);

    //@siclog "Step 24" siclog@
    //The UE transmits an RRCReestablishmentRequest message on Cell 2
    v_RNTI_Value := f_NR_CellInfo_GetRNTI(nr_Cell1);
    v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
    v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(nr_Cell2);
        
    f_NR_CellInfo_SetRNTI(nr_Cell2, tsc_C_RNTI_Value4);
    f_NR_SS_C_RNTI_Config(nr_Cell2, tsc_C_RNTI_Value4, -, tsc_NoCnfReq, tsc_FollowOnFlag);
    f_NR_SS_CommonCellConfig(nr_Cell2, cads_NR_RachProcedure_DcchDtchToCSS_Config_REQ(nr_Cell2, cs_TimingInfo_Now, tsc_CnfReq,
                                                                                      cs_NR_RachProcedureConfig_CBRA_Msg4Based(v_NR_PhysicalParameters, cs_NR_RachProcedureMsg4WithNoRrcMsg),
                                                                                      cs_NR_SearchSpaceDlDciAssignment_CSSType1(v_NR_PhysicalParameters),
                                                                                      cs_NR_SearchSpaceUlDciAssignmentCSS(v_NR_PhysicalParameters, cs_NR_DciFormat_0_0_Params)));
    v_ShortMAC_I := f_NR_Calculate_ShortMAC(nr_Cell1, nr_Cell2);
    SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell2, cr_38508_RRCReestablishmentRequest(cr_NR_ReestabUE_Identity (v_RNTI_Value, v_PhysCellId, v_ShortMAC_I), otherFailure)));

    //@siclog "Step 25 - 26 " siclog@
    <<SKIPPED CODE>>



After Change:
	  function f_TC_8_1_5_8_1_NR5GC_TestBody() runs on NR5GC_PTC
  {
    var RNTI_Value v_RNTI_Value;
    var PhysCellId v_PhysCellId;
    var ShortMAC_I v_ShortMAC_I;
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters;
    var NR_SecurityParams_Type v_SecurityParams;
    var octetstring v_EncodeCapRequestNR;
    var template (value) DRB_CountMSB_InfoList v_CountList;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var template (value) RadioBearerConfig v_RadioBearerConfigResume; //WA#WI=961473
    var template (value) CellGroupConfig v_CellGroupConfig;
    var NR_RadioBearerList_Type v_SS_Config;
    var template (value) RRCReconfiguration_v1530_IEs v_V1530Ext;
    var template (value) NR_RRC_MSG_Request_Type v_RrcMsg;
    var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
    var template(omit) octetstring v_EncodedSCG_Config := omit;
    var NR_DRBInfoList_Type v_ExistingDRBs;
    var template(value) NR_RachProcedureConfig_Type v_NR_RachProcedureConfig;
    var integer v_Max_nPRB := f_NR_CellInfo_GetCarrierBandwidth_DL(nr_Cell1);
    var NR_ResourceAllocation_Type v_NR_ResourceAllocation := {v_Max_nPRB, 28};
    var template DRB_CountInfoList v_CountList_UE;
    var template (value) NG_NAS_MSG_RequestList_Type v_ServerAcceptMsg;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template(value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    //WA#WI=961473_Start
    var template(omit) DRB_ToAddModList v_DRB_ToAddModList := omit;
    var integer i;
    timer t_Wait;
      
    

    v_CellPowerList_AtT0 :={
      cs_NR_CellPower(nr_Cell2, -96, -96) //FFS in prose for FR2
    };
    v_CellPowerList_AtT1:={
      cs_NR_CellPower(nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),   //FFS in prose for FR2
      cs_NR_CellPower(nr_Cell2, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)          //FFS in prose for FR2
    };
    f_NR_SetCellPowerList(v_CellPowerList_AtT0);

    //@siclog "Step 1" siclog@
    //The SS transmits a Paging message including a matched ng-5G-S-TMSI.
    f_NR_Paging_Def(nr_Cell1, RRC_IDLE);

    //@siclog "Step 2" siclog@
    //The UE transmits an RRCSetupRequest message.
    f_NR_RRC_SetupRequest_Def(nr_Cell1);

    //@siclog "Steps 3 - 4" siclog@
    //The SS transmits an RRCSetup message.
    //The SS starts transmitting a periodic DCI from (10*2^numerology+1+TDL) slots after step 3 to schedule PUSCH.
    //Check: Does the UE transmit an RRCSetupComplete message within (10*2^numerology+1+TDL+K2) slots15 (FDD)/18 (TDD) subframes after successful completion of step 3?
    v_ReceivedNR_SRB_COMMON_IND := f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                                                              cas_NR_SRB0_RrcPdu_REQ(nr_Cell1,cs_TimingInfo_Now,f_NR_38508_RRCSetup(nr_Cell1, tsc_NR_RRC_TI_Def)),
                                                              car_NR_SRB1_RrcNasPdu_IND(nr_Cell1, cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *), cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected)),
                                                              10, "Step 4" );

    //@siclog "Step 5 - 6" siclog@
    //The SS transmits a SecurityModeCommand message.
    //The SS starts transmitting a periodic DCI from (5*2^numerology+1+TDL) slots after step 5 to schedule PUSCH.
    //Check: Does the UE transmit a SecurityModeComplete message within (5*2^numerology+1+TDL+K2) 10 (FDD)/13 (TDD) subframes slots after successful completion of step 5?
    v_SecurityParams := f_NR_Security_Get();
    v_SecurityParams := f_NR_Authentication_InitAS (v_SecurityParams, v_ReceivedNR_SRB_COMMON_IND.Signalling.Nas[0].SecurityProtection.NasCount);
    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetInitial(nr_Cell1);
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(nr_Cell1,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);

    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_SRB1_RrcPdu_REQ(nr_Cell1,
                                                      -,
                                                      cs_38508_RRCSecurityModeCommand(tsc_NR_RRC_TI_Def,
                                                                                      cs_38508_SecurityAlgorithmConfig(v_SecurityParams.AS_Ciphering.Algorithm,
                                                                                                                       v_SecurityParams.AS_Integrity.Algorithm))),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCSecurityModeComplete(tsc_NR_RRC_TI_Def)),
                               5, "Step 6" );

    f_NR_Security_Set(v_SecurityParams);

    //@siclog "Step 7 - 8" siclog@
    //The SS transmits a UECapabilityEnquiry message.
    //The SS starts transmitting a periodic DCI from (80*2^numerology+1+TDL) slots after step 7 to schedule PUSCH.
    //Check: Does the UE transmit a UECapabilityInformation message within (80*2^numerology+1+TDL+K2) slots after successful completion of step 7?
    v_EncodeCapRequestNR := bit2oct(encvalue( cs_38508_UECapabilityRequestFilterNR({ cs_NR_BandInformationNR(f_NR_CellInfo_GetBandIndicator(nr_Cell1)) }) ));
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_UECapabilityEnquiry(tsc_NR_RRC_TI_Def, nr, v_EncodeCapRequestNR)),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_UECapabilityInformationAny(tsc_NR_RRC_TI_Def)),
                               80, "Step 8",
                               cs_NR_FreqDomainSchedulExplicit(-, v_NR_ResourceAllocation.Lrbs)  //DCI with the maximum grant allowed by the carrier bandwidth  @sic R5-217815 sic@
                               );

    //@siclog "Step 9 - 10" siclog@
    //The SS transmits an RRCReconfiguration message to establish SRB2 and DRB #n for the UE
    f_NR_BuildRRCReconfigExistingDRBs (v_RadioBearerConfig, v_CellGroupConfig, v_SS_Config, nr_Cell1);
    
    v_SecurityParams := f_NR_Security_Get();
    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(v_CellGroupConfig)));

    //Reconfigure DRB in SS - existing DRBs are not reconfigured during release
    f_NR_SS_CommonRadioBearerConfig(nr_Cell1, v_SS_Config, cs_TimingInfo_Now);
    
    // Configure PDCP security
    v_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(nr_Cell1, false);
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(nr_Cell1,
                                          v_SecurityParams.AS_Integrity,
                                          v_SecurityParams.AS_Ciphering);
    v_ServerAcceptMsg := {cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, f_Get_NG_ServiceAcceptMsg(v_ReceivedNR_SRB_COMMON_IND.Signalling.Nas[0].Pdu.Msg.service_Request.pduSessionStatus))};
    v_RrcMsg := cs_NR_RRC_MSG_Request_DCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig, -, -, v_V1530Ext));
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_RrcNasPduList_REQ(nr_Cell1, tsc_NR_RbId_SRB1, cs_TimingInfo_Now, v_RrcMsg, v_ServerAcceptMsg),
                               car_NR_SRB_RrcPdu_IND(nr_Cell1, tsc_NR_RbId_SRB1, cr_38508_RRCReconfigurationComplete),
                               10, "Step 10" );

    //@siclog "Step 11 - 12" siclog@
    //The SS transmits a CounterCheck message.
    //The SS starts transmitting a periodic DCI from (5*2^numerology+1+TDL) slots after step 11 to schedule PUSCH.
    //Check: Does the UE transmit a CounterCheckResponse message within (5*2^numerology+1+TDL+K2)10 (FDD)/13 (TDD) subframes slots after successful completion of step 11?
    v_ExistingDRBs := f_NR5GC_MobileInfo_GetDRBInfoList();
    v_CountList := {cs_38508_DRB_CountMSB_Info(v_ExistingDRBs[0].DRB_Id, 1, 1)};
    v_CountList_UE := {cr_DRB_CountInfo(v_ExistingDRBs[0].DRB_Id)};
    f_NR5GC_RRC_ProcedureDelay(nr_Cell1,
                               cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_38508_CounterCheck(tsc_NR_RRC_TI_Def, v_CountList)),
                               car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_CounterCheckResponse(tsc_NR_RRC_TI_Def, v_CountList_UE)),
                               5, "Step 12" );

    //@siclog "Step 13" siclog@
    //Void

    //@siclog "Step 14a1-14a7" siclog@
    if ( pc_inactiveState ){

        //@siclog "Step 14a1" siclog@
        //The SS transmits an RRCRelease message with suspendConfig.
        f_NR_RRC_InactiveDef(nr_Cell1);

        //@siclog "Step 14a2" siclog@
        //Wait 1s to let UE enter RRC_INACTIVE state.
        f_Delay(1.0);

        //@siclog "Step 14a3" siclog@
        //The SS transmits a Paging message including a matched ng-5G-S-TMSI.
        f_NR_Paging_Def(nr_Cell1, RRC_INACTIVE);

        v_CellGroupConfig := f_NR_GetSCGConfigResume(nr_Cell1);
        v_EncodedSCG_Config := bit2oct(encvalue(v_CellGroupConfig));
        
        //WA#WI=961473
        for (i:= 0; i < lengthof(v_ExistingDRBs); i :=i+1) {
            v_DRB_ToAddModList[i] := cs_NR_DRB_ToAddMod(omit, v_ExistingDRBs[i].DRB_Id, true_, -, omit);
        }
        v_RadioBearerConfigResume := cs_38508_RadioBearerConfigDef( {cs_NR_SRB_ToAddMod_ReestablishPDCP(tsc_SRB2)}, v_DRB_ToAddModList);

        //@siclog "Step 14a4" siclog@
        //The UE transmits an RRCResumeRequest message
        SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeRequest(tsc_NR_ShortI_RNTI_Value1, mt_Access)));

        //@siclog "Step 14a5" siclog@
        //The SS transmits an RRCResume message for the UE

        if(pc_reducedCP_Latency){
            //@siclog "Step 14a5Aa1 - 14a5Aa2" siclog@
            //The SS starts transmitting a periodic DCI from (10* 2 2^numerology +1+TDL) slots after step 14a5 to schedule PUSCH.
            //Check: Does the UE transmit an RRCResumeComplete message within (6*2^numerology+1+TDL+K2) slots after successful completion of step 14a5?
            f_NR5GC_RRC_ProcedureDelay( nr_Cell1,
                                        cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, cs_38508_RRCResume(tsc_NR_RRC_TI_Def, v_EncodedSCG_Config, v_RadioBearerConfigResume)), //WA#WI=961473
                                        car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeComplete),
                                        6, "Step 14a5Aa2" );
        } else {
            //@siclog "Step 14a5Ab1 - 14a5Ab2" siclog@
            //The SS starts transmitting a periodic DCI from n slots after step 14a5 to schedule PUSCH.
            //Check: Does the UE transmit an RRCResumeComplete message within (10* 2 2^numerology +1+TDL+K2)slots after successful completion of step 14a5?
            f_NR5GC_RRC_ProcedureDelay( nr_Cell1,
                                        cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, cs_TimingInfo_Now, cs_38508_RRCResume(tsc_NR_RRC_TI_Def, v_EncodedSCG_Config, v_RadioBearerConfigResume)), //WA#WI=961473
                                        car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCResumeComplete),
                                        10, "Step 14a5Ab2" );
        }
        v_NR_RachProcedureConfig := f_NR_InitialiseRachProcedure(nr_Cell1);
        f_NR_SS_RachProcedureConfig(nr_Cell1, v_NR_RachProcedureConfig);

        //@siclog "Step 14a6-14a7" siclog@
        //Void
    }
    
    //@siclog "Step 14-21" siclog@
    //Void

    //@siclog "Step 22" siclog@
    //The SS re-adjusts the SS levels according to row "T1" in table 8.1.5.8.1.3.2-1/2 in order that the radio link quality of NR Cell 1 is degraded, and NR Cell 2 is suitable for camping
    f_NR_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 23" siclog@
    //Wait 1s to let T310 expire
    //f_Delay(1.0); //WA#WI=961473
    t_Wait.start (1.0); //WA#WI=961473

//@siclog "Step 24" siclog@
    //The UE transmits an RRCReestablishmentRequest message on Cell 2
   v_RNTI_Value := f_NR_CellInfo_GetRNTI(nr_Cell1);
    v_PhysCellId := f_NR_CellInfo_GetPhysicalCellId(nr_Cell1);
    v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(nr_Cell2);
        
    f_NR_CellInfo_SetRNTI(nr_Cell2, tsc_C_RNTI_Value4);
    f_NR_SS_C_RNTI_Config(nr_Cell2, tsc_C_RNTI_Value4, -, tsc_NoCnfReq, tsc_FollowOnFlag);
    f_NR_SS_CommonCellConfig(nr_Cell2, cads_NR_RachProcedure_DcchDtchToCSS_Config_REQ(nr_Cell2, cs_TimingInfo_Now, tsc_CnfReq,
                                                                                      cs_NR_RachProcedureConfig_CBRA_Msg4Based(v_NR_PhysicalParameters, cs_NR_RachProcedureMsg4WithNoRrcMsg),
                                                                                      cs_NR_SearchSpaceDlDciAssignment_CSSType1(v_NR_PhysicalParameters),
                                                                                      cs_NR_SearchSpaceUlDciAssignmentCSS(v_NR_PhysicalParameters, cs_NR_DciFormat_0_0_Params)));
    v_ShortMAC_I := f_NR_Calculate_ShortMAC(nr_Cell1, nr_Cell2);
    //WA#WI=961473
    alt{ 
        []SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell2, cr_38508_RRCReestablishmentRequest(cr_NR_ReestabUE_Identity (v_RNTI_Value, v_PhysCellId, v_ShortMAC_I), otherFailure)))
        {
                t_Wait.stop;
        }
        []t_Wait.timeout
        {
                 SRB.receive(car_NR_SRB0_RrcPdu_IND(nr_Cell2, cr_38508_RRCReestablishmentRequest(cr_NR_ReestabUE_Identity (v_RNTI_Value, v_PhysCellId, v_ShortMAC_I), otherFailure)));
        }
    }   

    //@siclog "Step 25 - 26 " siclog@
    <<SKIPPED CODE>>


0.2 [bookmark: _Toc103669276]Correction to the function f_NR5GC_RRC_ProcedureDelay
	Function name
	f_NR5GC_RRC_ProcedureDelay

	Reason for change
	In the implementation of function f_NR5GC_RRC_ProcedureDelay :
Disabling of UL HARQ System Indications are done once HARQ ACK is received . there are possibilities of race condition where TTCN is waiting for  the CNF for Disable UL HARQ System indication , but ended up receiving SRB COMMON IND. This race condition has to be addressed

	Summary of change
	To avoid the race condition between the reception of SYSTEM CONTROL CNF for Disbaling UL HARQ indications and SRB COMMON IND , new handler is added inside the alt block

	TTCN module
	\Common\NR5GC\NR5GC_RRCSteps.ttcn

	MCC160 Comment
	



    
Before Change:
	function f_NR5GC_RRC_ProcedureDelay(NR_CellId_Type                        p_NR_CellId,
                                      template (value) NR_SRB_COMMON_REQ    p_SRB_COMMON_REQ,
                                      template (present) NR_SRB_COMMON_IND  p_SRB_COMMON_IND,
                                      integer                               p_Nms,
                                      charstring                            p_TestcaseStep,
                                      template (value) NR_FreqDomainSchedulExplicit_Type p_FreqDomainSchedul := cs_NR_FreqDomainSchedulExplicitUL_Def
                                      ) runs on NR_BASE_PTC return NR_SRB_COMMON_IND
  {
    var integer v_T1Slots;
    var integer v_T2Slots;
    var integer v_T3Slots;
    var integer v_T4Slots;
    var SubFrameTiming_Type v_T1; // Send DL Msg
    var SubFrameTiming_Type v_T2; // HARQ ACK receive
    var SubFrameTiming_Type v_T3; // Start UL Grant start
    var SubFrameTiming_Type v_T4; // UL RRC PDU received
    var SubFrameTiming_Type v_Tnow;
    var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
    var integer v_Nslots := p_Nms*v_SlotsNumberPerSubframe + 1;  //@sic R5s210842 change 2.1 sic@
    var integer v_Delta1;
    var integer v_Delta2;
    var integer v_GrantPeriodicity;
    var integer v_DL_RetransmissionCnt := 0;
    var NR_SYSTEM_IND v_SYS_IND;
    var boolean v_T2Flag := false;
    var boolean v_T4Flag := false;
    var integer v_K1;
    var integer v_K2 := 4;
    var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
    var integer v_MaxNumberSlots := 1024 * 10 * v_SlotsNumberPerSubframe;     // Maximum number per SFN complete cycle
    var template (value) NR_SRB_COMMON_REQ v_SRB_COMMON_REQ := p_SRB_COMMON_REQ;

    timer t_Watchdog;
    
    //@sic R5s210842 sic@
    if (f_TimingInfo_IsNow(v_SRB_COMMON_REQ.Common.TimingInfo)) {
        v_T1 := f_NR_GetNextSendOccasion_DL(p_NR_CellId);
        v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame := v_T1;
    } else {
        v_T1 := valueof(v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame);
    }
    v_T1Slots := f_NR_ConvertTiming_To_TimingInSlots(v_T1, v_SlotsNumberPerSubframe);
    // Initialise GrantPeriodicity, Delta1, K1
    if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
        v_GrantPeriodicity := 1; // Grant repeated in every slot
        v_Delta1 := 0;
        v_K1 := 7;
    } else { // TDD
        select(v_SCS) {
            case(kHz15) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5 - (v_T1Slots + v_Nslots)) mod 5;
                v_K1 := 7;
            }
            case(kHz30) {
                v_GrantPeriodicity := 10; //Grant repeated every 10 slots
                v_Delta1 := (4 - (v_T1Slots + v_Nslots) mod 10) mod 10;
                v_K1 := 5;
            }
            case(kHz120) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5- (v_T1Slots + v_Nslots) mod 5) mod 5;
                v_K1 := 5;
            }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    }
    
    //Calculate T3
    v_T3Slots := v_T1Slots + v_Nslots + v_Delta1;  //@sic R5s210842 change 2 sic@
    v_T3 := f_NR_ConvertTimingInSlots_To_Timing(v_T3Slots, v_SCS, v_SlotsNumberPerSubframe);
    
    //T1: send DL Msg
    SRB.send(v_SRB_COMMON_REQ);

    //Enable HARQ ack/nack reporting
    f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(enable), cs_TimingInfo_NR(v_T1)); //@sic R5s210842 change 2.4 sic@
    
    //T3 configure SS with periodic UL Grant at T3
    f_NR_ULGrantConfiguration_Start(p_NR_CellId, cs_TimingInfo_NR(v_T3), -, p_FreqDomainSchedul, -, cs_UL_GrantConfig_ContPeriodic(v_GrantPeriodicity));
                                          
    // Start WatchDog timer
    v_Tnow := f_NR_GetCurrentTiming(p_NR_CellId);
    t_Watchdog.start (int2float(f_SubFrameTiming_Duration(v_Tnow, v_T1) + p_Nms + 50) / 1000.0);  /* 50ms added to be at the safe side */
    
    alt {
        /* HARQ NACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, nack)) {
            v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;
            if (v_DL_RetransmissionCnt > 4) {
                t_Watchdog.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "DL Msg retransmissions > 4");
            } else {
                repeat;
            }
        }
        /* HARQ ACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, ack))-> value v_SYS_IND {
            v_T2 := v_SYS_IND.Common.TimingInfo.SubFrame;
            f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(disable));    /* disable indication of HARQ ack/nack; */
            v_T2Flag := true;
            repeat;
        }
        /* HARQ error indication */
        [] SYSIND.receive (car_NR_SYSTEM_HARQ_ERROR(p_NR_CellId, cr_NR_HarqErrorUL_Any)) {
            f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in RRC Delay test");
            t_Watchdog.stop;
        }
        [] SRB.receive(p_SRB_COMMON_IND) -> value v_ReceivedNR_SRB_COMMON_IND {
            v_T4 := v_ReceivedNR_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
            v_T4Flag := true;
            t_Watchdog.stop;
        }
        [] t_Watchdog.timeout {                                                                                     /* timeout => after T5 */
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "t_Watchdog timeout");
        }
    }
    <<SKIPPED CODE>>



After Change:
	function f_NR5GC_RRC_ProcedureDelay(NR_CellId_Type                        p_NR_CellId,
                                      template (value) NR_SRB_COMMON_REQ    p_SRB_COMMON_REQ,
                                      template (present) NR_SRB_COMMON_IND  p_SRB_COMMON_IND,
                                      integer                               p_Nms,
                                      charstring                            p_TestcaseStep,
                                      template (value) NR_FreqDomainSchedulExplicit_Type p_FreqDomainSchedul := cs_NR_FreqDomainSchedulExplicitUL_Def
                                      ) runs on NR_BASE_PTC return NR_SRB_COMMON_IND
  {
    var integer v_T1Slots;
    var integer v_T2Slots;
    var integer v_T3Slots;
    var integer v_T4Slots;
    var SubFrameTiming_Type v_T1; // Send DL Msg
    var SubFrameTiming_Type v_T2; // HARQ ACK receive
    var SubFrameTiming_Type v_T3; // Start UL Grant start
    var SubFrameTiming_Type v_T4; // UL RRC PDU received
    var SubFrameTiming_Type v_Tnow;
    var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
    var integer v_Nslots := p_Nms*v_SlotsNumberPerSubframe + 1;  //@sic R5s210842 change 2.1 sic@
    var integer v_Delta1;
    var integer v_Delta2;
    var integer v_GrantPeriodicity;
    var integer v_DL_RetransmissionCnt := 0;
    var NR_SYSTEM_IND v_SYS_IND;
    var boolean v_T2Flag := false;
    var boolean v_T4Flag := false;
    var boolean v_SysIndCnfFlag := false; //WA#WI=961473
    var integer v_K1;
    var integer v_K2 := 4;
    var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
    var integer v_MaxNumberSlots := 1024 * 10 * v_SlotsNumberPerSubframe;     // Maximum number per SFN complete cycle
    var template (value) NR_SRB_COMMON_REQ v_SRB_COMMON_REQ := p_SRB_COMMON_REQ;

    timer t_Watchdog;
    
    //@sic R5s210842 sic@
    if (f_TimingInfo_IsNow(v_SRB_COMMON_REQ.Common.TimingInfo)) {
        v_T1 := f_NR_GetNextSendOccasion_DL(p_NR_CellId);
        v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame := v_T1;
    } else {
        v_T1 := valueof(v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame);
    }
    v_T1Slots := f_NR_ConvertTiming_To_TimingInSlots(v_T1, v_SlotsNumberPerSubframe);
    // Initialise GrantPeriodicity, Delta1, K1
    if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
        v_GrantPeriodicity := 1; // Grant repeated in every slot
        v_Delta1 := 0;
        v_K1 := 7;
    } else { // TDD
        select(v_SCS) {
            case(kHz15) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5 - (v_T1Slots + v_Nslots)) mod 5;
                v_K1 := 7;
            }
            case(kHz30) {
                v_GrantPeriodicity := 10; //Grant repeated every 10 slots
                v_Delta1 := (4 - (v_T1Slots + v_Nslots) mod 10) mod 10;
                v_K1 := 5;
            }
            case(kHz120) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5- (v_T1Slots + v_Nslots) mod 5) mod 5;
                v_K1 := 5;
            }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    }
    
    //Calculate T3
    v_T3Slots := v_T1Slots + v_Nslots + v_Delta1;  //@sic R5s210842 change 2 sic@
    v_T3 := f_NR_ConvertTimingInSlots_To_Timing(v_T3Slots, v_SCS, v_SlotsNumberPerSubframe);
    
    //T1: send DL Msg
    SRB.send(v_SRB_COMMON_REQ);

    //Enable HARQ ack/nack reporting
    f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(enable), cs_TimingInfo_NR(v_T1)); //@sic R5s210842 change 2.4 sic@
    
    //T3 configure SS with periodic UL Grant at T3
    f_NR_ULGrantConfiguration_Start(p_NR_CellId, cs_TimingInfo_NR(v_T3), -, p_FreqDomainSchedul, -, cs_UL_GrantConfig_ContPeriodic(v_GrantPeriodicity));
                                          
    // Start WatchDog timer
    v_Tnow := f_NR_GetCurrentTiming(p_NR_CellId);
    t_Watchdog.start (int2float(f_SubFrameTiming_Duration(v_Tnow, v_T1) + p_Nms + 50) / 1000.0);  /* 50ms added to be at the safe side */
    
    alt {
        /* HARQ NACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, nack)) {
            v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;
            if (v_DL_RetransmissionCnt > 4) {
                t_Watchdog.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "DL Msg retransmissions > 4");
            } else {
                repeat;
            }
        }
        /* HARQ ACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, ack))-> value v_SYS_IND {
            v_T2 := v_SYS_IND.Common.TimingInfo.SubFrame;
            //f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(disable));    /* disable indication of HARQ ack/nack; */ //WA#WI=961473
            SYS.send(cas_NR_SystemIndCtrl_REQ(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(disable))); //WA#WI=961473
            v_T2Flag := true;
            repeat;
        } //WA#WI=961473
        [] SYS.receive(car_NR_SystemIndCtrl_CNF(p_NR_CellId)) //WA#WI=961473
        {
            v_SysIndCnfFlag := true;
            if(not v_T4Flag){repeat;}                
        }
        /* HARQ error indication */
        [] SYSIND.receive (car_NR_SYSTEM_HARQ_ERROR(p_NR_CellId, cr_NR_HarqErrorUL_Any)) {
            f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in RRC Delay test");
            t_Watchdog.stop;
        }
        [] SRB.receive(p_SRB_COMMON_IND) -> value v_ReceivedNR_SRB_COMMON_IND {            
            v_T4 := v_ReceivedNR_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
            v_T4Flag := true;
            t_Watchdog.stop;
            if(not v_T2Flag)  //WA#WI=961473
            {
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL ACK not received");
            };
            if(not v_SysIndCnfFlag){repeat;}  //WA#WI=961473
        }
        [] t_Watchdog.timeout {                                                                                     /* timeout => after T5 */
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "t_Watchdog timeout");
        }
    } 
    <<SKIPPED CODE>>


0.3 [bookmark: _Toc103669277]Correction to the function f_NR5GC_RRC_ProcedureDelayReestablishment
	Function name
	f_NR5GC_RRC_ProcedureDelayReestablishment

	Reason for change
	In the implementation of function f_NR5GC_RRC_ProcedureDelayReestablishment
1.Disabling of UL HARQ System Indications are done once HARQ ACK is received . there are possibilities of race condition where TTCN is waiting for  the CNF for Disable UL HARQ System indication , but ended up receiving SRB COMMON IND. In Rach case , when Disabling Prach System Indications are done , once Prach Preamble is received , there are possibilities of race condition where TTCN is waiting for the CNF for Disable Prach System Indication , but ended up receiving SRB COMMON IND.
This race condition has to be addressed

2.when UE has triggered RACH procedure for sending the reestablishment complete message , “repeat” is missing inside the alt block . this results in SRB COMMON IND ( Reestablishment complete ) reception failure

3. v_T1Slots calculcation is not considering v_MaxNumberSlots


	Summary of change
	1.To avoid the race condition between the reception of SYSTEM CONTROL CNF for Disabling UL HARQ indications and SRB COMMON IND , new handler is added inside the alt block ; same is taken care between SYSTEM CONTROL CNF for Disabling prach Indications and SRB COMMON IND

In the function f_NR5GC_RRC_ProcedureDelayReestablishment
2.repeat statement is added inside alt block , when receiving the Rach Preamble Indication , so that Reestablishment complete gets handled inside the same alt block

3. MaxNumberSlots is taken into consideration while calculating  v_T1Slots

	TTCN module
	\Common\NR5GC\NR5GC_RRCSteps.ttcn

	MCC160 Comment
	



    
Before Change:
	function f_NR5GC_RRC_ProcedureDelayReestablishment(NR_CellId_Type                        p_NR_CellId,
                                                     template (value) NR_SRB_COMMON_REQ    p_SRB_COMMON_REQ,
                                                     template (present) NR_SRB_COMMON_IND  p_SRB_COMMON_IND,
                                                     integer                               p_Nms,
                                                     charstring                            p_TestcaseStep
                                                    ) runs on NR_BASE_PTC return NR_SRB_COMMON_IND
  {
    var integer v_T1Slots;
    var integer v_T2Slots;
    var integer v_T3Slots;
    var integer v_T4Slots;
    var SubFrameTiming_Type v_T1; // Send DL Msg
    var SubFrameTiming_Type v_T2; // HARQ ACK receive
    var SubFrameTiming_Type v_T3; // Start UL Grant start
    var SubFrameTiming_Type v_T4; // UL RRC PDU received
    var SubFrameTiming_Type v_Tnow;
    var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
    var integer v_Nslots := p_Nms*v_SlotsNumberPerSubframe + 1;  //@sic R5s210842 change 2.1 sic@
    var integer v_Delta1;
    var integer v_Delta2;
    var integer v_GrantPeriodicity;
    var integer v_DL_RetransmissionCnt := 0;
    var NR_SYSTEM_IND v_SYS_IND;
    var boolean v_T2Flag := false;
    var boolean v_T4Flag := false;
    var integer v_K1;
    var integer v_K2 := 4;
    var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
    var integer v_MaxNumberSlots := 1024 * 10 * v_SlotsNumberPerSubframe;     // Maximum number per SFN complete cycle
    var boolean v_RACH_ProcedureStarted := false;
    var integer v_DeltaTDLmax := 0;
    var template (value) NR_SRB_COMMON_REQ v_SRB_COMMON_REQ := p_SRB_COMMON_REQ;
    
    timer t_Watchdog;

    // The SS transmits an RRCReestablishment message on slot #i in the radio frame,where:
    // i = 1 for micro=0,1
    // i = 21 for micro=3
    v_T1 := f_NR_GetNextSendOccasion_DL(p_NR_CellId);
    v_T1.SFN.Number := (v_T1.SFN.Number + 1) mod 1024;
    select (v_SCS) {
        case (kHz15) {
            v_T1.Subframe.Number := 1;
            v_T1.Slot.SlotOffset.Numerology0 := true;
        }
        case (kHz30) {
            v_T1.Subframe.Number := 0;
            v_T1.Slot.SlotOffset.Numerology1 := 1;
        }
        case (kHz120) {
            v_T1.Subframe.Number := 2;
            v_T1.Slot.SlotOffset.Numerology3 := 5;
        }
        case else { FatalError(__FILE__, __LINE__, "Invalid SCS");}
    }
    v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame := v_T1;

    v_T1Slots := f_NR_ConvertTiming_To_TimingInSlots(v_T1, v_SlotsNumberPerSubframe);
    // Initialise GrantPeriodicity, Delta1, K1
    if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
        v_GrantPeriodicity := 1; // Grant repeated in every slot
        v_Delta1 := 0;
        v_K1 := 7;
        select (v_SCS) {
            case (kHz15) { v_DeltaTDLmax := 13; }
            case (kHz30) { v_DeltaTDLmax := 30; }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    } else { // TDD
        select(v_SCS) {
            case(kHz15) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5 - (v_T1Slots + v_Nslots)) mod 5;
                v_K1 := 7;
                v_DeltaTDLmax := 7;
            }
            case(kHz30) {
                v_GrantPeriodicity := 10; //Grant repeated every 10 slots
                v_Delta1 := (4 - (v_T1Slots + v_Nslots) mod 10) mod 10;
                v_K1 := 5;
                v_DeltaTDLmax := 17;
            }
            case(kHz120) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5- (v_T1Slots + v_Nslots) mod 5) mod 5;
                v_K1 := 5;
                v_DeltaTDLmax := 32;
            }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    }
    
    //Calculate T3 - UL grant when CSS is used
    v_T3Slots := v_T1Slots + v_Nslots + v_Delta1;
    v_T3 := f_NR_ConvertTimingInSlots_To_Timing(v_T3Slots, v_SCS, v_SlotsNumberPerSubframe);
    
    //T1: send DL Msg
    SRB.send(v_SRB_COMMON_REQ);

    //Enable HARQ ack/nack reporting
    f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(enable), cs_TimingInfo_NR(v_T1)); //@sic R5s210842 change 2.4 sic@
    //Enable RACH preamble reporting
    f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_RachPreamble(enable));
    
    //T3 configure SS with periodic UL Grant at T3
    f_NR_ULGrantConfiguration_Start(p_NR_CellId, cs_TimingInfo_NR(v_T3), -, -, -, cs_UL_GrantConfig_ContPeriodic(v_GrantPeriodicity));
                                          
    // Start WatchDog timer
    v_Tnow := f_NR_GetCurrentTiming(p_NR_CellId);
    t_Watchdog.start (int2float(f_SubFrameTiming_Duration(v_Tnow, v_T1) + p_Nms + 50) / 1000.0);  /* 50ms added to be at the safe side */
    
    alt {
        /* HARQ NACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, nack)) {
            v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;
            if (v_DL_RetransmissionCnt > 4) {
                t_Watchdog.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "DL Msg retransmissions > 4");
            } else {
                repeat;
            }
        }
        /* HARQ ACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, ack))-> value v_SYS_IND {
            v_T2 := v_SYS_IND.Common.TimingInfo.SubFrame;
            f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(disable));    /* disable indication of HARQ ack/nack; */
            v_T2Flag := true;
            repeat;
        }
        /* HARQ error indication */
        [] SYSIND.receive (car_NR_SYSTEM_HARQ_ERROR(p_NR_CellId, cr_NR_HarqErrorUL_Any)) {
            f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in RRC Delay test");
            t_Watchdog.stop;
        }
        [] SYSIND.receive (car_NR_SYSTEM_HARQ_ERROR(p_NR_CellId, cr_NR_HarqErrorUL_Any)) {
            f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in RRC Delay test");
            t_Watchdog.stop;
        }
        /* UE initiates RACH procedure */
        [] SYSIND.receive(car_NR_PRACH_Preamble_IND(?, cr_TimingInfo_Any, ?)) {
            f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_RachPreamble(disable));
            v_RACH_ProcedureStarted := true;
        }
        [] SRB.receive(p_SRB_COMMON_IND) -> value v_ReceivedNR_SRB_COMMON_IND {
            v_T4 := v_ReceivedNR_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
            v_T4Flag := true;
            t_Watchdog.stop;
        }
        [] t_Watchdog.timeout {  /* timeout => after T5 */
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "t_Watchdog timeout");
        }
    }
    //Stop UL Grant
    f_NR_ULGrantConfiguration_Stop(p_NR_CellId);
    
    if (not v_T2Flag) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL ACK not received");
    }
    if (not v_T4Flag) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL Msg not received");
    }
    
    v_T4Slots := (((v_T4.SFN.Number * 10) + v_T4.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T4) + v_MaxNumberSlots;
    if (not v_RACH_ProcedureStarted) {
    // No RACH procedure was initiated by UE
    // Timing calculation as per 38.523-3 clause 7.3.7.2
        //Check the Delay
        v_T2Slots := (((v_T2.SFN.Number * 10) + v_T2.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T2) + v_MaxNumberSlots;
        v_T4Slots := (((v_T4.SFN.Number * 10) + v_T4.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T4) + v_MaxNumberSlots;
        //Initialise Delta2
        if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
            v_Delta2 := 0;
        } else { //TDD
            select(v_SCS) {
                case(kHz15) {
                    v_Delta2 := (5 - (v_T2Slots - v_K1 + v_Nslots)) mod 5;
                }
                case(kHz30) {
                    v_Delta2 := (4 - (v_T2Slots - v_K1 + v_Nslots) mod 10) mod 10;
                }
                case(kHz120) {
                    v_Delta2 := (5- (v_T2Slots - v_K1 + v_Nslots) mod 5) mod 5;
                }
                case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
            }
        }
        if ( ((v_T4Slots - v_K2) - (v_T2Slots - v_K1))mod v_MaxNumberSlots <= (v_Nslots + v_Delta2)) {
            f_NR_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);
        } else {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);
        }
    } else {
    // RACH procedure was initiated by UE
    // Timing calculation as per 38.523-3 clause 7.3.7.3
        v_T1Slots := (((v_T1.SFN.Number * 10) + v_T1.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T1);
        if ( (v_T4Slots - v_T1Slots) <= ( v_Nslots + v_DeltaTDLmax) ) {
            f_NR_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);
        } else {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);
        }
   }
   return v_ReceivedNR_SRB_COMMON_IND;
  }



After Change:
	function f_NR5GC_RRC_ProcedureDelayReestablishment(NR_CellId_Type                        p_NR_CellId,
                                                     template (value) NR_SRB_COMMON_REQ    p_SRB_COMMON_REQ,
                                                     template (present) NR_SRB_COMMON_IND  p_SRB_COMMON_IND,
                                                     integer                               p_Nms,
                                                     charstring                            p_TestcaseStep
                                                    ) runs on NR_BASE_PTC return NR_SRB_COMMON_IND
  {
    var integer v_T1Slots;
    var integer v_T2Slots;
    var integer v_T3Slots;
    var integer v_T4Slots;
    var SubFrameTiming_Type v_T1; // Send DL Msg
    var SubFrameTiming_Type v_T2; // HARQ ACK receive
    var SubFrameTiming_Type v_T3; // Start UL Grant start
    var SubFrameTiming_Type v_T4; // UL RRC PDU received
    var SubFrameTiming_Type v_Tnow;
    var integer v_SlotsNumberPerSubframe := f_NR_CellInfo_GetSlotsNumberPerSubframe(p_NR_CellId);
    var SubcarrierSpacing v_SCS := f_NR_CellInfo_GetSCS(p_NR_CellId);
    var integer v_Nslots := p_Nms*v_SlotsNumberPerSubframe + 1;  //@sic R5s210842 change 2.1 sic@
    var integer v_Delta1;
    var integer v_Delta2;
    var integer v_GrantPeriodicity;
    var integer v_DL_RetransmissionCnt := 0;
    var NR_SYSTEM_IND v_SYS_IND;
    var boolean v_T2Flag := false;
    var boolean v_T4Flag := false;
    var boolean v_SysIndCnfFlagHarq := false; //WA#WI=961473
    var boolean v_SysIndCnfFlagRach := false; //WA#WI=961473
    var integer v_K1;
    var integer v_K2 := 4;
    var NR_SRB_COMMON_IND v_ReceivedNR_SRB_COMMON_IND;
    var integer v_MaxNumberSlots := 1024 * 10 * v_SlotsNumberPerSubframe;     // Maximum number per SFN complete cycle
    var boolean v_RACH_ProcedureStarted := false;
    var integer v_DeltaTDLmax := 0;
    var template (value) NR_SRB_COMMON_REQ v_SRB_COMMON_REQ := p_SRB_COMMON_REQ;
    
    
    timer t_Watchdog;

    // The SS transmits an RRCReestablishment message on slot #i in the radio frame,where:
    // i = 1 for micro=0,1
    // i = 21 for micro=3
    v_T1 := f_NR_GetNextSendOccasion_DL(p_NR_CellId);
    v_T1.SFN.Number := (v_T1.SFN.Number + 1) mod 1024;
    select (v_SCS) {
        case (kHz15) {
            v_T1.Subframe.Number := 1;
            v_T1.Slot.SlotOffset.Numerology0 := true;
        }
        case (kHz30) {
            v_T1.Subframe.Number := 0;
            v_T1.Slot.SlotOffset.Numerology1 := 1;
        }
        case (kHz120) {
            v_T1.Subframe.Number := 2;
            v_T1.Slot.SlotOffset.Numerology3 := 5;
        }
        case else { FatalError(__FILE__, __LINE__, "Invalid SCS");}
    }
    v_SRB_COMMON_REQ.Common.TimingInfo.SubFrame := v_T1;

    v_T1Slots := f_NR_ConvertTiming_To_TimingInSlots(v_T1, v_SlotsNumberPerSubframe);
    // Initialise GrantPeriodicity, Delta1, K1
    if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
        v_GrantPeriodicity := 1; // Grant repeated in every slot
        v_Delta1 := 0;
        v_K1 := 7;
        select (v_SCS) {
            case (kHz15) { v_DeltaTDLmax := 13; }
            case (kHz30) { v_DeltaTDLmax := 30; }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    } else { // TDD
        select(v_SCS) {
            case(kHz15) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5 - (v_T1Slots + v_Nslots)) mod 5;
                v_K1 := 7;
                v_DeltaTDLmax := 7;
            }
            case(kHz30) {
                v_GrantPeriodicity := 10; //Grant repeated every 10 slots
                v_Delta1 := (4 - (v_T1Slots + v_Nslots) mod 10) mod 10;
                v_K1 := 5;
                v_DeltaTDLmax := 17;
            }
            case(kHz120) {
                v_GrantPeriodicity := 5; //Grant repeated every 5 slots
                v_Delta1 := (5- (v_T1Slots + v_Nslots) mod 5) mod 5;
                v_K1 := 5;
                v_DeltaTDLmax := 32;
            }
            case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
        }
    }
    
    //Calculate T3 - UL grant when CSS is used
    v_T3Slots := v_T1Slots + v_Nslots + v_Delta1;
    v_T3 := f_NR_ConvertTimingInSlots_To_Timing(v_T3Slots, v_SCS, v_SlotsNumberPerSubframe);
      
    //T1: send DL Msg
    SRB.send(v_SRB_COMMON_REQ);

    //Enable HARQ ack/nack reporting
    f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(enable), cs_TimingInfo_NR(v_T1)); //@sic R5s210842 change 2.4 sic@
    //Enable RACH preamble reporting
    f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_RachPreamble(enable));
    
    //T3 configure SS with periodic UL Grant at T3
    f_NR_ULGrantConfiguration_Start(p_NR_CellId, cs_TimingInfo_NR(v_T3), -, -, -, cs_UL_GrantConfig_ContPeriodic(v_GrantPeriodicity));
                                          
    // Start WatchDog timer
    v_Tnow := f_NR_GetCurrentTiming(p_NR_CellId);
    
    t_Watchdog.start (int2float(f_SubFrameTiming_Duration(v_Tnow, v_T1) + p_Nms + 50) / 1000.0);  /* 50ms added to be at the safe side */
    
    alt {
        /* HARQ NACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, nack)) {
            v_DL_RetransmissionCnt := v_DL_RetransmissionCnt + 1;
            if (v_DL_RetransmissionCnt > 4) {
                t_Watchdog.stop;
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "DL Msg retransmissions > 4");
            } else {
                repeat;
            }
        }
        /* HARQ ACK */
        []SYSIND.receive(car_NR_SYSTEM_UL_HARQ_IND(p_NR_CellId, ack))-> value v_SYS_IND {
            v_T2 := v_SYS_IND.Common.TimingInfo.SubFrame;
            SYS.send(cas_NR_SystemIndCtrl_REQ(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(disable))); //WA#WI=961473
            //f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_UL_HARQ(disable));    /* disable indication of HARQ ack/nack; */ //WA#WI=961473
            v_T2Flag := true;
            repeat;
        } //WA#WI=961473
        [] SYS.receive(car_NR_SystemIndCtrl_CNF(p_NR_CellId)) //WA#WI=961473
        { //WA#WI=961473
            v_SysIndCnfFlagHarq := true;
            if(v_RACH_ProcedureStarted){v_SysIndCnfFlagRach := true;}
            if((not v_T4Flag) or (v_RACH_ProcedureStarted and not v_SysIndCnfFlagRach)){repeat;}                
        }  
        /* HARQ error indication */
        [] SYSIND.receive (car_NR_SYSTEM_HARQ_ERROR(p_NR_CellId, cr_NR_HarqErrorUL_Any)) {
            f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in RRC Delay test");
            t_Watchdog.stop;
        }
        [] SYSIND.receive (car_NR_SYSTEM_HARQ_ERROR(p_NR_CellId, cr_NR_HarqErrorUL_Any)) {
            f_SetVerdictInconc(__FILE__, __LINE__, p_TestcaseStep & "HARQ error for UL transmission in RRC Delay test");
            t_Watchdog.stop;
        }
        /* UE initiates RACH procedure */
        [] SYSIND.receive(car_NR_PRACH_Preamble_IND(?, cr_TimingInfo_Any, ?)) {
            //f_NR_SS_SystemIndCtrlConfig(p_NR_CellId, cds_NR_SystemIndCtrl_RachPreamble(disable)); //WA#WI=961473
            SYS.send(cas_NR_SystemIndCtrl_REQ(p_NR_CellId, cds_NR_SystemIndCtrl_RachPreamble(disable))); //WA#WI=961473
            v_RACH_ProcedureStarted := true;
            repeat; //WA#WI=961473
        }
        [] SRB.receive(p_SRB_COMMON_IND) -> value v_ReceivedNR_SRB_COMMON_IND {
            v_T4 := v_ReceivedNR_SRB_COMMON_IND.Common.TimingInfo.SubFrame;
            v_T4Flag := true;
            t_Watchdog.stop;
            if(not v_T2Flag)  //WA#WI=961473
            {
                f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL ACK not received");
            };
            if((not v_SysIndCnfFlagHarq) or (v_RACH_ProcedureStarted and not v_SysIndCnfFlagRach)){repeat;}  //WA#WI=961473
        }
        [] t_Watchdog.timeout {  /* timeout => after T5 */
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "t_Watchdog timeout");
        }
    }
    //Stop UL Grant
    f_NR_ULGrantConfiguration_Stop(p_NR_CellId);
    
    if (not v_T2Flag) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL ACK not received");
    }
    if (not v_T4Flag) {
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep & "UL Msg not received");
    }
    
    v_T4Slots := (((v_T4.SFN.Number * 10) + v_T4.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T4) + v_MaxNumberSlots;
    if (not v_RACH_ProcedureStarted) {
    // No RACH procedure was initiated by UE
    // Timing calculation as per 38.523-3 clause 7.3.7.2
        //Check the Delay
        v_T2Slots := (((v_T2.SFN.Number * 10) + v_T2.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T2) + v_MaxNumberSlots;
        v_T4Slots := (((v_T4.SFN.Number * 10) + v_T4.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T4) + v_MaxNumberSlots;
        //Initialise Delta2
        if (f_NR_CellInfo_GetIsFDD(p_NR_CellId)){
            v_Delta2 := 0;
        } else { //TDD
            select(v_SCS) {
                case(kHz15) {
                    v_Delta2 := (5 - (v_T2Slots - v_K1 + v_Nslots)) mod 5;
                }
                case(kHz30) {
                    v_Delta2 := (4 - (v_T2Slots - v_K1 + v_Nslots) mod 10) mod 10;
                }
                case(kHz120) {
                    v_Delta2 := (5- (v_T2Slots - v_K1 + v_Nslots) mod 5) mod 5;
                }
                case else { FatalError(__FILE__, __LINE__, "Invalid SCS"); }
            }
        }
        
        if ( ((v_T4Slots - v_K2) - (v_T2Slots - v_K1))mod v_MaxNumberSlots <= (v_Nslots + v_Delta2)) {
            f_NR_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);
        } else {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);
        }
    } else {
        // RACH procedure was initiated by UE
        // Timing calculation as per 38.523-3 clause 7.3.7.3
        v_T1Slots := (((v_T1.SFN.Number * 10) + v_T1.Subframe.Number) * v_SlotsNumberPerSubframe) + f_NR_SubFrameTiming_GetSlotNumber(v_T1) + v_MaxNumberSlots; //WA#WI=961473
      
        if ( (v_T4Slots - v_T1Slots) <= ( v_Nslots + v_DeltaTDLmax) ) {
        f_NR_PreliminaryPass(__FILE__, __LINE__, p_TestcaseStep);
    } else {
            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, p_TestcaseStep);
    }
   }
   return v_ReceivedNR_SRB_COMMON_IND;
  }




